




**Just do It:**

# **Simple Monadic Equational Reasoning**

***Jeremy Gibbons (jww Ralf Hinze)***

***Shonan, March 2019***

# 1. Reasoning with effects?



## 2. Monads in Haskell

An interface for effectful computation:

```
class Monad m where
  return :: a → m a
  (≫=)   :: m a → (a → m b) → m b
```

Unit and associativity laws:

```
return x ≫= k   = k x
mx ≫= return   = mx
(mx ≫= k) ≫= k' = mx ≫= (λx → k x ≫= k')
```

Two abbreviations:

```
skip :: Monad m ⇒ m ()      (≫)      :: Monad m ⇒ m a → m b → m b
skip = return ()           mx ≫ my = mx ≫= const my
```

## 2.1. Imperative functional programming

**do** { *e* } = *e*  
**do** { *x* ← *e*; *es* } = *e*  $\gg$   $\lambda x \rightarrow$  **do** { *es* }  
**do** { *e*; *es* } = *e*  $\gg$  **do** { *es* }  
**do** { **let** *decls*; *es* } = **let** *decls* **in** **do** { *es* }

‘Haskell is the world’s best imperative programming language.’ (SPJ)

### 3. A counter example

The *Monad* interface provides general-purpose plumbing.  
For any particular class of effect, we need additional operations.

```
class Monad m ⇒ MonadCount m where  
  tick :: m ()
```

Then, for example, Towers of Hanoi:

```
hanoi :: MonadCount m ⇒ Int → m ()  
hanoi 0      = do { skip }  
hanoi (n + 1) = do { hanoi n; tick; hanoi n }
```



## 3.1. Correctness

We claim that

$$\text{hanoi } n = \mathbf{do} \{ \text{rep } (2^n - 1) \text{ tick} \}$$

where

$$\text{rep} :: \text{Monad } m \Rightarrow \text{Int} \rightarrow m () \rightarrow m ()$$
$$\text{rep } 0 \quad mx = \mathbf{do} \{ \text{skip} \}$$
$$\text{rep } (n + 1) \text{ mx} = \mathbf{do} \{ \text{mx}; \text{rep } n \text{ mx} \}$$

Note that

$$\text{rep } 1 \quad mx = \mathbf{do} \{ \text{mx} \}$$
$$\text{rep } (m + n) \text{ mx} = \mathbf{do} \{ \text{rep } m \text{ mx}; \text{rep } n \text{ mx} \}$$

## 3.2. Reasoning

Proof by induction. Base case trivial:

$$\mathit{hanoi} \ 0 = \mathbf{do} \ \{\mathit{skip}\} = \mathbf{do} \ \{\mathit{rep} \ (2^0 - 1) \ \mathit{tick}\}$$

For inductive step,

$$\begin{aligned} & \mathit{hanoi} \ (n + 1) \\ = & \quad \llbracket \text{definition of } \mathit{hanoi} \ \rrbracket \\ & \mathbf{do} \ \{\mathit{hanoi} \ n; \mathit{tick}; \mathit{hanoi} \ n\} \\ = & \quad \llbracket \text{inductive hypothesis; } \mathit{rep} \ 1 \ \rrbracket \\ & \mathbf{do} \ \{\mathit{rep} \ (2^n - 1) \ \mathit{tick}; \mathit{rep} \ 1 \ \mathit{tick}; \mathit{rep} \ (2^n - 1) \ \mathit{tick}\} \\ = & \quad \llbracket \mathit{rep} \ \text{promotes through addition} \ \rrbracket \\ & \mathbf{do} \ \{\mathit{rep} \ (2^n - 1 + 1 + 2^n - 1) \ \mathit{tick}\} \\ = & \quad \llbracket \text{arithmetic} \ \rrbracket \\ & \mathbf{do} \ \{\mathit{rep} \ (2^{n+1} - 1) \ \mathit{tick}\} \end{aligned}$$

A particularly simple example, because *MonadCount* algebra is free.

## 4. Failure, choice and nondeterminism

A class of possibly failing computations:

```
class Monad m ⇒ MonadFail m where  
  fail :: m a
```

such that *fail* is a left zero of sequential composition:

$$\text{fail} \gg m = \text{fail}$$

(but not a right zero!).

Useful shorthand:

```
guard :: MonadFail m ⇒ Bool → m ()  
guard b = if b then skip else fail
```



## 4.1. Choice

A class of computations that make choices:

**class** *MonadAlt* *m* **where**

$(\square) :: m\ a \rightarrow m\ a \rightarrow m\ a$

such that  $\square$  is associative, and composition distributes leftwards over it:

$(m \square n) \square p = m \square (n \square p)$

$(m \square n) \gg= k = (m \gg= k) \square (n \gg= k)$

(but not rightwards!).

## 4.2. Nondeterminism

... as a combination of failure and choice:

**class** (*MonadFail m, MonadAlt m*)  $\Rightarrow$  *MonadNondet m*

No additional operations. But two additional unit laws:

*fail*  $\square$  *mx* = *mx* = *mx*  $\square$  *fail*

Finite lists, bags, and sets are instances

(the latter two adding commutativity and idempotence of ( $\square$ ), respectively).

## 4.3. Permutations

For example,

```
perms :: MonadNondet m => [a] -> m [a]
perms [] = do { return [] }
perms xs = do { (y, ys) ← select xs; zs ← perms ys; return (y : zs) }
```

where

```
select :: MonadNondet m => [a] -> m (a, [a])
select [] = do { fail }
select (x : xs) = do { return (x, xs) } □
                    do { (y, ys) ← select xs; return (y, x : ys) }
```

## 5. State

A class of computations exploiting updatable state:


```
class Monad m  $\Rightarrow$  MonadState s m | m  $\rightarrow$  s where  
  get :: m s  
  put :: s  $\rightarrow$  m ()
```

with four axioms:

```
put s  $\gg$  put s'           = put s'  
put s  $\gg$  get                = put s  $\gg$  return s  
get  $\gg$  put                   = skip  
get  $\gg$   $\lambda s \rightarrow$  get  $\gg$  k s = get  $\gg$   $\lambda s \rightarrow$  k s s
```

## 5.1. Eight queens

Queen at  $(r, c)$  threatens up-diagonal  $r-c$  and down-diagonal  $r+c$ :



The essence of queen safety:

$$\text{test} :: (\text{Int}, \text{Int}) \rightarrow ([\text{Int}], [\text{Int}]) \rightarrow (\text{Bool}, ([\text{Int}], [\text{Int}])))$$

$$\text{test } (c, r) \text{ (ups, downs)} = (u \notin \text{ups} \wedge d \notin \text{downs}, (u : \text{ups}, d : \text{downs}))$$

where  $(u, d) = (r-c, r+c)$

## 5.2. Eight queens, purely

The safety test for a candidate layout:

$$\begin{aligned}
 \text{safe}_1 &:: ([Int], [Int]) \rightarrow [(Int, Int)] \rightarrow (Bool, ([Int], [Int])) \\
 \text{safe}_1 &= \text{foldr } \text{step}_1 \circ \text{start}_1 \text{ where} \\
 &\quad \text{start}_1 \text{ updowns} = (True, \text{updowns}) \\
 &\quad \text{step}_1 \text{ cr } (\text{restOK}, \text{updowns}) = (\text{thisOK} \wedge \text{restOK}, \text{updowns}') \\
 &\quad \text{where } (\text{thisOK}, \text{updowns}') = \text{test } \text{cr } \text{updowns}
 \end{aligned}$$

Then generate and test:

$$\begin{aligned}
 \text{queens} &:: \text{MonadNondet } m \Rightarrow \text{Int} \rightarrow m [Int] \\
 \text{queens } n &= \mathbf{do} \{ rs \leftarrow \text{perms } [1..n]; \\
 &\quad \text{guard } (\text{fst } (\text{safe}_1 \text{ empty } (\text{place } n \text{ rs}))); \text{return } rs \} \\
 \text{place } n \text{ rs} &= \text{zip } [1..n] \text{ rs} \\
 \text{empty} &= ([], [])
 \end{aligned}$$

## 5.3. Safety testing, statefully

Maintain the checked diagonals statefully:

```

safe2 :: MonadState ([Int], [Int]) m => [(Int, Int)] → m Bool
safe2 = foldr step2 start2 where
  start2      = do { return True }
  step2 cr k = do { restOK ← k; updowns ← get;
                    let (thisOK, updowns') = test cr updowns;
                        put updowns'; return (thisOK ∧ restOK) }

```

Simple proof using axioms of *get* and *put* that

```

safe2 crs
= do { updowns ← get;
      let (ok, updowns') = safe1 updowns crs;
          put updowns'; return ok }

```

## 6. Combining effects

Nondeterminism for permutations, state for safety testing:

**class** (*MonadState s m, MonadNondet m*)  $\Rightarrow$  *MonadStateNondet s m* |  $m \rightarrow s$

Again, no new operations, but some additional laws—*fail* also a right zero:

$m \gg \text{fail} = \text{fail}$

and composition distributes also rightwards over choice:

$m \gg= \lambda x \rightarrow k_1 x \square k_2 x = (m \gg= k_1) \square (m \gg= k_2)$

That is, *local* or *backtrackable state*. (Each choice point entails a clean slate.)

In particular, guards commute with anything:

$\text{guard } b \gg m = m \gg= \lambda x \rightarrow \text{guard } b \gg \text{return } x$



## 6.1. Queens, nondeterministically and statefully

Using  $get \gg= put = skip$  and commuting guards, calculate

$$\begin{aligned}
 \text{queens } n &= \mathbf{do} \{ rs \leftarrow \text{perms } [1..n]; \\
 &\quad \text{guard } (\text{fst } (\text{safe}_1 \text{ empty } (\text{place } n \text{ rs}))); \text{return } rs \} \\
 &= \mathbf{do} \{ s \leftarrow \text{get}; rs \leftarrow \text{perms } [1..n]; \text{put empty}; \\
 &\quad \text{ok} \leftarrow \text{safe}_2 (\text{place } n \text{ rs}); \text{put } s; \text{guard ok}; \text{return } rs \} \\
 &= \mathbf{do} \{ s \leftarrow \text{get}; rs \leftarrow \text{perms } [1..n]; \text{put empty}; \\
 &\quad \text{ok} \leftarrow \text{safe}_2 (\text{place } n \text{ rs}); \text{guard ok}; \text{put } s; \text{return } rs \} \\
 &= \mathbf{do} \{ s \leftarrow \text{get}; rs \leftarrow \text{perms } [1..n]; \text{put empty}; \\
 &\quad \text{safe}_3 (\text{place } n \text{ rs}); \text{put } s; \text{return } rs \}
 \end{aligned}$$

where  $\text{safe}_3 \text{ crs} = \text{safe}_2 \text{ crs} \gg= \text{guard}$ . Then calculate that

$$\text{safe}_3 \text{ crs} = \text{foldr } \text{step}_3 \text{ start}_3 \mathbf{where} \text{step}_3 = \dots; \text{start}_3 = \dots$$

by plain ordinary equational reasoning.

## 7. Think locally, act globally

- state and failure combine, in two ways
- *local* state:  $s \rightarrow \text{Maybe } (a, s)$
- *global* state:  $s \rightarrow (\text{Maybe } a, s)$
- different interactions between the two theories
- state and nondeterminism combine nicely *locally*:  $s \rightarrow [(a, s)]$
- but sometimes you want *global* state
- eg Prolog evaluator, or playing Sudoku
- however,  $s \rightarrow ([a], s)$  is not a monad
- what is the equational theory? and implementation?

8	5	9	<del>6</del>	4	7	2	1	3
2	1	4	<del>9</del>	3	8	7	5	<del>6</del>
<del>7</del>	6	<del>3</del>	2	5	1	8	9	4
<del>5</del>	9	7	4	6	2	1	<del>3</del>	8
<del>6</del>	2	8	<del>1</del>	9	<del>3</del>	5	4	<del>7</del>
<del>3</del>	4	1	<del>8</del>	7	<del>5</del>	9	<del>6</del>	2
<del>1</del>	<del>7</del>	6	5	8	4	3	2	9
<del>4</del>	8	2	<del>3</del>	1	<del>9</del>	6	<del>7</del>	5
<del>9</del>	3	5	<del>7</del>	2	6	4	<del>8</del>	1

## 8. Summary

- computational effects as algebraic theories
- the axioms are important! as with type classes etc too
- theories combine—trivially, or with interaction
- *making equations great again*
- personal bugbear:  
*language designers are  
compiler writers*
- *Just do it,*  
JG and Ralf Hinze,  
ICFP 2011

